

# Solving Matrix Equations

Worksheet by Michael K. May S.J., revised by Russell Blyth

```
> with(LinearAlgebra):
```

## Outline:

The basic objectives are:

- 1) Learn to use Maple to multiply matrices and vectors.
- 2) Learn to use Maple to solve Matrix-vector equations.
- 3) To complete an extended exercise.

```
>
```

## Part 1: Multiplying a matrix and a vector

We want to change our point of view. So far we have been focused on solving systems of linear equations and have manipulated matrices to perform that task. In this section we consider the equivalent matrix equation and how to solve it directly. Before we can talk about solving matrix equations we need to be able to multiply a matrix and an appropriately sized vector. We start by defining a matrix  $A$  and a vector  $X$  of appropriate size so that the product  $AX$  is defined. (The product  $AX$  of an  $m$  by  $n$  matrix  $A$  with the vector  $X$  is only defined if  $X$  is an  $n$ -vector.)

```
> A := Matrix(3,4,[[1, 2, 1, 3], [-5, 7, 2, 2],[ 13, 4, 4, 3]]);  
X := Vector(4, [1, 3, -2, 4]);
```

$$A := \begin{bmatrix} 1 & 2 & 1 & 3 \\ -5 & 7 & 2 & 2 \\ 13 & 4 & 4 & 3 \end{bmatrix}$$

$$X := \begin{bmatrix} 1 \\ 3 \\ -2 \\ 4 \end{bmatrix}$$

(2.1)

Recall that we can create the same matrix and vector using the angle bracket constructor.

```
> A1 := <<1, -5, 13> | <2, 7, 4> | <1, 2, 4> | <3, 2, 3>>;  
X1 := <1, 3, -2, 4>;
```

$$A1 := \begin{bmatrix} 1 & 2 & 1 & 3 \\ -5 & 7 & 2 & 2 \\ 13 & 4 & 4 & 3 \end{bmatrix}$$

(2.2)

$$XI := \begin{bmatrix} 1 \\ 3 \\ -2 \\ 4 \end{bmatrix} \quad (2.2)$$

Now we want to compute the product. Maple uses `.` as the symbol for the noncommutative product of matrices.

```
> B := A.X;
```

$$B := \begin{bmatrix} 17 \\ 20 \\ 29 \end{bmatrix} \quad (2.3)$$

Maple also lets you use commands for the operations. The command to multiply a matrix followed by a vector is `MatrixVectorMultiply`. There are similar commands for a vector and a matrix as well as for the product of two matrices.

```
> MatrixVectorMultiply(A,X);
```

$$\begin{bmatrix} 17 \\ 20 \\ 29 \end{bmatrix} \quad (2.4)$$

Recall that the product  $AX$  can be thought of as a linear combination of the columns of  $A$  with the scalars used being the entries of  $X$ . If we compute this linear combination, we should get the same result.

```
> Column(A,1)*X[1]+Column(A,2)*X[2]+Column(A,3)*X[3]+Column(A,4)*X[4];
```

$$\begin{bmatrix} 17 \\ 20 \\ 29 \end{bmatrix} \quad (2.5)$$

Note that multiplication of a matrix by a vector is not commutative. In fact in our case,  $X.A$  is not even defined.

```
> X.A;
```

```
Error, (in LinearAlgebra:-VectorMatrixMultiply) invalid input:
LinearAlgebra:-VectorMatrixMultiply expects its 1st argument, v,
to be of type Vector[row] but received Vector[Column]( 1..4, [...
unable to display content ...] )
```

```
> VectorMatrixMultiply(X,A);
```

```
Error, (in LinearAlgebra:-VectorMatrixMultiply) invalid input:
LinearAlgebra:-VectorMatrixMultiply expects its 1st argument, v,
to be of type Vector[row] but received Vector[Column]( 1..4, [...
unable to display content ...] )
```

Sometimes we want to put two matrices (or a matrix and a vector) together into a larger matrix. We can do this with the angle bracket constructor, using a comma to put matrices on top of each other and a vertical bar to put them next to each other.

```
> <A|B>;
<A,Transpose(X)>;
```

$$\begin{bmatrix} 1 & 2 & 1 & 3 & 17 \\ -5 & 7 & 2 & 2 & 20 \\ 13 & 4 & 4 & 3 & 29 \\ 1 & 2 & 1 & 3 & 17 \\ -5 & 7 & 2 & 2 & 20 \\ 13 & 4 & 4 & 3 & 29 \\ 1 & 3 & -2 & 4 & \end{bmatrix}$$

(2.6)

### Exercises:

1) Use the commands `RandomMatrix` and `RandomVector` to create a random 4 by 5 matrix named `Mat1` and a 5-vector named `Vec1`. Compute the product `Mat1 . Vec1`, assigned to a 4-vector named `Vec2`.

```
>
```

2) Augment `Mat1` by `Vec2` to create a 4 by 6 matrix. (Read carefully:) Stack `Vec1` on top of `Mat1` to create a 5 by 5 matrix.

```
>
```

```
>
```

### Part 2: Solving a matrix-vector equation

```
>
```

Having computed  $B = AX$ , we want to shift the focus: given the matrix  $A$  and the vector  $B$ , solve for  $X$  in the matrix equation  $AX = B$ .

The first method is to make an augmented matrix from  $A$  and  $B$ , then use Gaussian elimination to produce a matrix in echelon form. We then use back substitution to find the general solution to the matrix equation.

```
> C1 := <A|B>;
   C1a := GaussianElimination(C1);
   gensol1 := BackwardSubstitute(C1a);
```

$$C1 := \begin{bmatrix} 1 & 2 & 1 & 3 & 17 \\ -5 & 7 & 2 & 2 & 20 \\ 13 & 4 & 4 & 3 & 29 \end{bmatrix}$$

$$C1a := \begin{bmatrix} 1 & 2 & 1 & 3 & 17 \\ 0 & 17 & 7 & 17 & 105 \\ 0 & 0 & \frac{1}{17} & -14 & \frac{-954}{17} \end{bmatrix}$$

(3.1)

$$\text{gensol1} := \begin{bmatrix} 173 - 43 \_t_1 \\ 399 - 99 \_t_1 \\ -954 + 238 \_t_1 \\ -t_1 \end{bmatrix} \quad (3.1)$$

Here  $\_t[1]$  is a parameter that can take any value. To get back to our original X, we need to set  $\_t[1]$  equal to 4.

> **eval(gensol1, \\_t[1]=4);**

$$\begin{bmatrix} 1 \\ 3 \\ -2 \\ 4 \end{bmatrix} \quad (3.2)$$

A slight variation of the method above is to perform Gauss-Jordan elimination before back substitution.

> **C1b := ReducedRowEchelonForm(C1);**  
**gensol2 := BackwardSubstitute(C1b);**

$$C1b := \begin{bmatrix} 1 & 0 & 0 & 43 & 173 \\ 0 & 1 & 0 & 99 & 399 \\ 0 & 0 & 1 & -238 & -954 \end{bmatrix}$$

$$\text{gensol2} := \begin{bmatrix} 173 - 43 \_t_1 \\ 399 - 99 \_t_1 \\ -954 + 238 \_t_1 \\ -t_1 \end{bmatrix} \quad (3.3)$$

A third option is to use the `linsolve` command that we saw in the previous worksheet.

> **gensol3 := LinearSolve(A,B);**

$$\text{gensol3} := \begin{bmatrix} 173 - 43 \_t_4 \\ 399 - 99 \_t_4 \\ -954 + 238 \_t_4 \\ -t_4 \end{bmatrix} \quad (3.4)$$

A fourth method notes that the general solution to the equation  $AX = B$  is any vector of the form  $t_v + n_v$ , where  $t_v$  is a particular solution to the equation (a translation vector), and  $n_v$  is any vector in the null space of A. The Maple command `nullspace(A)` returns a basis for the nullspace of A. (A basis of a vector space is a spanning set of minimal size.)

> **NullSpace(A);**

$$\left\{ \begin{bmatrix} -43 \\ -99 \\ 238 \\ 1 \end{bmatrix} \right\} \quad (3.5)$$

### Exercises:

- 3) Find a basis of the null space of the matrix Mat1 you created above.  
 >  
 4) Solve the matrix-vector equation (Mat1)(X) = Vec2 by reducing an augmented matrix.  
 >  
 5) Solve the matrix-vector equation (Mat1)(X) = Vec2 by using the linsolve command.  
 >  
 >

### Part 3: An Extended exercise:

For these exercises we will work with a fixed matrix A and a fixed vector B.

$$\text{Let } A = \begin{bmatrix} 17 & -6 & 13 & 27 & 64 & 19 \\ 4 & -6 & -33 & 25 & 7 & 9 \\ 55 & -24 & 6 & 106 & 199 & 66 \\ 89 & -36 & 32 & 160 & 327 & 104 \end{bmatrix} \text{ and } B = \begin{bmatrix} 17 \\ 4 \\ 55 \\ 89 \end{bmatrix}.$$

#### Exercises:

- 6) Compute the rank of A and determine the number of free variables in the system AX = 0.

```
> A := Matrix(<< 17 | -6 | 13 | 27 | 64 | 19 >,
              < 4 | -6 | -33 | 25 | 7 | 9 >,
              < 55 | -24 | 6 | 106 | 199 | 66 >,
              < 89 | -36 | 32 | 160 | 327 | 104 >>);
```

$$A := \begin{bmatrix} 17 & -6 & 13 & 27 & 64 & 19 \\ 4 & -6 & -33 & 25 & 7 & 9 \\ 55 & -24 & 6 & 106 & 199 & 66 \\ 89 & -36 & 32 & 160 & 327 & 104 \end{bmatrix} \quad (4.1)$$

Find a basis for the nullspace of A. (Name the vectors of your basis W1, W2, ...) How could you have predicted the size of the basis?

```
> W:=NullSpace(A);
W1:= W[1]; W2:= W[2]; W3 := W[3]; W4 := W[4];
```

$$W := \left\{ \begin{bmatrix} \frac{-2}{13} \\ \frac{317}{78} \\ 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} \frac{-46}{13} \\ \frac{-613}{78} \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} \frac{-10}{13} \\ \frac{77}{78} \\ 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} \frac{-57}{13} \\ \frac{-137}{78} \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix} \right\}$$

$$W1 := \begin{bmatrix} \frac{-2}{13} \\ \frac{317}{78} \\ 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}$$

$$W2 := \begin{bmatrix} \frac{-57}{13} \\ \frac{-137}{78} \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}$$

$$W3 := \begin{bmatrix} \frac{-46}{13} \\ \frac{-613}{78} \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

$$W4 := \begin{bmatrix} -\frac{10}{13} \\ \frac{77}{78} \\ 0 \\ 0 \\ 0 \\ 1 \end{bmatrix} \quad (4.2)$$

By inspection, find a vector  $X0$  such that  $(A)(X0)=B$ .

>  $x0 := <1,0,0,0,0,0>;$   
 $A.x0;$

$$X0 := \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$


$$\begin{bmatrix} 17 \\ 4 \\ 55 \\ 89 \end{bmatrix}$$

(4.3)

Show that any vector of the form  $X0 + a1*W1 + a2*W2 + \dots$  is a solution to the equation  $AX = B$ .

>  $x := x0 + a1*W1 + a2*W2 + a3*W3 + a4*W4;$   
 $A.x;$

$$X := \begin{bmatrix} 1 - \frac{2}{13} a1 - \frac{57}{13} a2 - \frac{46}{13} a3 - \frac{10}{13} a4 \\ \frac{317}{78} a1 - \frac{137}{78} a2 - \frac{613}{78} a3 + \frac{77}{78} a4 \\ a3 \\ a1 \\ a2 \\ a4 \end{bmatrix}$$



$$\begin{bmatrix} 17 \\ 4 \\ 55 \\ 89 \end{bmatrix}$$

(4.4)